

1982/4. A FORTRAN program for the drift correction of gravity data
(Revision 2)

R.G. Richardson

Abstract

This program removes a linear drift from gravity readings made at times between the occupation of pairs of base stations. The base stations need not be the same.

THE PROGRAM

GRVDFT (Appendix 1)

Earlier versions of this program (Richardson and Leaman, 1981; Richardson, 1981) calculated drift between a single pair of base stations. The program now allows drift calculations between a succession of base stations and prints a record of all data entry and gravity differences.

The program reads station numbers, station times, and meter readings on logical unit 5 and prints drift and datum corrected differences on logical units 6 and 7. All times are in 24 hour notation, with a decimal point separating the hours and minutes (e.g. 10.25). Non-standard sub-routine calls are:

CALL SYSCOM (I, 'AS 0622, 0720, 0899*') sets the logical units for the printer, optional output, and the core buffer

CALL SYSCOM (I, 'AS 0700*') sends the optional output to the null device

CALL SYSCOM (I, 'AS 0620, 0720, 0820*') resets the logical units to the system console.

Station numbers are read in free integer format using subroutine INFREE, and all other data is read in F10.1 format. Data should be input in the order first base station, second base station, stations between these two base stations, third base station, stations between the second and third base stations, fourth base station, data between the third and fourth base stations, etc.

The data for each base station is:

SNUM - the base station number
TIME - the time of reading the base station
BASE - the meter reading at the base station
SHIFT - any shift to be applied to the base value to correct it to the datum (in the same units as BASE). If the station number is the same as that of the last base station, SHIFT is not required.

The data for each station is:

SNUM - the station number. If SNUM = 0 the program asks for another base station. If there is no other base station, accumulated data is then printed and the program stops.
STIME - the time of reading the station
STNVAL - the station reading (in the same units as BASE)

Output is:

Logical Unit 6: The base station data, and then for each station the base station number (reset to one at the start of a run) from which the difference has been calculated, the drift (and datum) corrected difference, the reading time, and the uncorrected reading.

Logical Unit 7: The drift (and datum) corrected difference. This output appears only if the user requests the differences to be printed immediately.

REFERENCES

RICHARDSON, R.G.; LEAMAN, D.E. 1981. FORTRAN program for the drift correction of gravity data. *Unpubl.Rep.Dep.Mines Tasm.* 1981/37.

RICHARDSON, R.G. 1981. A revised FORTRAN program for the drift correction of gravity data. *Unpubl.Rep.Dep.Mines Tasm.* 1981/49.

[19 February 1982]

APPENDIX 1
Program GRVDFT

```

$TITL GRAVITY DRIFT CORRECTION
C TITLE GRVDFT
C A PROGRAM TO REMOVE A LINEAR DRIFT FROM GRAVITY READINGS
C RELATIVE TO A SERIES OF BASE STATIONS WHICH MAY OR MAY NOT BE THE SAME
C AND MAY OR MAY NOT BE THE SURVEY DATUM.
C TIMES MUST BE IN 24 HOUR NOTATION
C OUTPUT TO THE PRINTER IS STORED IN ARRAYS AND PRINTED AT THE END OF
C THE RUN
      INTEGER*2 SNUM(150), FFEED, BNUM(20), BSNUM(150), K, AOK, YES, NO, I
      IMPLICIT INTEGER*2 (I-N)
      LOGICAL IER
      REAL TIME1, TIME2, BTIME(20), BBASE(20), BSHIFT(20), DRIFT, STIME,
      STNVAL(150), VALUE(150), ATIME(150), ATIME1, ATIME2
      DATA YES/1HY/, NO/1HN/, FFEED/3072/
      K=1
      JBASE=1
      CALL SYSCOM(I, 'AS 0622, 0720, 0899*')
C SET LOGICAL UNIT 6 TO PRINTER, 7 TO INPUT DEVICE, 8 TO CORE BUFFER
      WRITE(5, 110)
110  FORMAT(' PRINT DIFFERENCES IMMEDIATELY?')
      READ(5, 111) AOK
111  FORMAT(A1)
      IF (AOK .NE. YES) CALL SYSCOM(I, 'AS 0700*')
C IF OUTPUT NOT WANTED SET LOGICAL UNIT 7 TO A NULL DEVICE
      WRITE(5, 96)
      96  FORMAT(' BASE 1 NUMBER?')
      CALL INFREE(BNUM(JBASE), IER, 5)
      55  WRITE(5, 100)
100  FORMAT(' BASE 1 TIME?')
      READ(5, 101) TIME1
101  FORMAT(F10.1)
C READ TIME AS E.G. 10.56 FOR TEN FIFTY-SIX
C MUST BE IN 24 HOUR FORMAT
      IF (TIME1 .GT. 24.0) GOTO 55
C NO DECIMAL POINT IN TIME1
      BTIME(JBASE)=TIME1
C SAVE FOR LATER PRINT
      CALL HOURS(TIME1)
C CORRECTION TO HOURS
      WRITE(5, 102)
102  FORMAT(' BASE 1 READING?')
      READ(5, 101) BBASE(JBASE)
      WRITE(5, 103)
103  FORMAT(' BASE 1 SHIFT?')
      READ(5, 101) BSHIFT(JBASE)
      50  JBSL1=JBASE
          JBASE=JBASE+1
          WRITE(5, 97) JBASE
      97  FORMAT(' BASE ', I2, ' NUMBER?')
      CALL INFREE(BNUM(JBASE), IER, 5)
      56  WRITE(5, 104) JBASE
104  FORMAT(' BASE ', I2, ' TIME?')
      READ(5, 101) TIME2
C MUST BE IN 24 HOUR FORMAT
      IF (TIME2 .GT. 24.0) GOTO 56
C TIME HAS NO DEC. PT.
      BTIME(JBASE)=TIME2
C SAVE FOR LATER PRINT
      CALL HOURS(TIME2)
      WRITE(5, 105) JBASE

```

```

105 FORMAT(' BASE ',I2,' READING?')
  READ(5,101) BBASE(JBASE)
  IF (BNUM(JBSL1) .NE. BNUM(JBASE)) GOTO 30
  BSHIFT(JBASE)=BSHIFT(JBSL1)
C IF THE SAME BASE STATION JUST COPY THE SHIFT
  GOTO 31
  30 WRITE(5,106) JBASE
106 FORMAT(' BASE ',I2,' SHIFT?')
  READ(5,101) BSHIFT(JBASE)
C HAVE READ THE BASE STATION PARAMETERS
C
C NOW CALCULATE THE DRIFT PER HOUR
  31 DTEMP=BBASE(JBSL1)-BSHIFT(JBSL1)
  DRIFT=((BBASE(JBASE)-BSHIFT(JBASE))-DTEMP)/(TIME2-TIME1)
C
  9 WRITE(5,98)
  98 FORMAT(' STN. NO. ?')
  CALL INFREE(SNUM(K),IER,5)
  IF (SNUM(K) .EQ. 0) GOTO 20
C TERMINATE INTERVAL BETWEEN BASES BY 0 STATION NUMBER
  10 WRITE(5,107)
107 FORMAT(' STN TIME?')
  READ(5,101) STIME
  IF (STIME .GT. 24.0) GOTO 10
C TIME HAS NO DEC. PT.
  ATIME(K)=STIME
C STORE THE TIME FOR THE PRINT AT THE END
  CALL HOURS(STIME)
  WRITE(5,108)
108 FORMAT(' STN VALUE?')
  READ(5,101) STNVAL(K)
  BSNUM(K)=JBSL1
C REMEMBER WHICH BASE THE DIFF IS CALCULATED RELATIVE TO
C
C NOW CALCULATE DRIFT CORRECTED VALUE
  VALUE(K)=STNVAL(K)-(DRIFT*(STIME-TIME1))-DTEMP
  WRITE(7,109) VALUE(K)
109 FORMAT(' VALUE',F9.1)
  K=K+1
C OK SO INCREMENT COUNTER BY ONE
  GOTO 9
  20 WRITE(5,112)
112 FORMAT(' ANOTHER BASE STATION?')
  READ(5,111) AOK
  IF (AOK .EQ. NO) GOTO 22
  IF (AOK .NE. YES) GOTO 20
  TIME1=TIME2
  GOTO 50
  22 K=K-1
  IF (K .EQ. 0) GOTO 21
C NO STATIONS
  WRITE(6,200) FFEED
200 FORMAT(A1,' BASE  SNUM    TIME      OBS      SHIFT')
  WRITE(6,203) (I,BNUM(I),BTIME(I),BBASE(I),BSHIFT(I),I=1,JBASE)
203 FORMAT(I5,I6,F6.2,2F9.1)
  WRITE(6,204)
204 FORMAT('/' BASE  SNUM      DIFF      TIME      OBS')
  WRITE(6,201) (BSNUM(I),SNUM(I),VALUE(I),ATIME(I),STNVAL(I),I=1,K)
201 FORMAT(I5,I6,F9.1,F9.2,F9.1)
  WRITE(6,202)

```

```
202 FORMAT(////)
21 CALL SYSCOM(I, 'AS 0620,0720,0820*')
C RE-ASSIGN ALL LOGICAL UNITS TO THE INPUT DEVICE
STOP
END
SUBROUTINE HOURS(TIME)
REAL HR, MINS, TIME
HR=IFIX(TIME+0.005)
C ENSURE TIME ROUNDS DOWN TO THE CORRECT HOUR
MINS=TIME-HR
C MINUTES
MINS=MINS*1.666667
C CONVERT MINUTES TO DECIMAL HOURS
TIME=HR+MINS
RETURN
END
```